

## **ТИПОВАЯ МЕТОДИКА И ИНСТРУМЕНТАРИЙ КОГНИТИВНОЙ СТРУКТУРИЗАЦИИ И ФОРМАЛИЗАЦИИ ЗАДАЧ В СК-АНАЛИЗЕ**

Луценко Е.В. – д. э. н., профессор

Кубанский государственный аграрный университет

В статье рассматриваются два варианта конкретной реализации первого этапа системно-когнитивного анализа (СК-анализ), т.е. когнитивная структуризация и формализация предметной области, отличающиеся формой представления исходных данных: 1) файл типа "Текст-DOS"; 2) таблицы Word или Excel. Для первого варианта приводится исходный текст универсального программного интерфейса, полностью автоматизирующего первый этап СК-анализа, для второго – полный инструментарий, основанный на использовании стандартных возможностей Excel.

### **1. Формулирование проблемы**

На практике часто встречаются задачи, в которых исследование объекта познания даже в принципе не может быть осуществлено на основе проведения с ним каких-либо экспериментов. В этом случае приходится использовать ретроспективные данные об этом объекте и изучение реального объекта в различных условиях заменить исследованием его формальной модели. При этом результаты, полученные путем моделирования, могут считаться изучением самого объекта только тогда, когда модель адекватна.

Метод системно-когнитивного анализа (СК-анализ) включает следующие этапы [1]:

1. Когнитивная структуризация, а затем формализация предметной области.

2. Ввод данных мониторинга в базу прецедентов за период, в течение которого имеется необходимая информация в электронной форме.

3. Синтез семантической информационной модели (СИМ).

4. Оптимизация СИМ.

5. Проверка адекватности СИМ (измерение внутренней и внешней, дифференциальной и интегральной валидности).

6. Анализ СИМ.

7. Решение задач идентификации состояний объекта управления, прогнозирование и поддержка принятия управленческих решений с применением СИМ.

Однако прежде чем осуществить синтез и исследование модели объекта управления, необходимо:

– во-первых, разработать способ преобразования исходных данных из формы, в которой они находятся, в форму, пригодную для обработки на компьютере;

– во-вторых, осуществить это преобразование на практике.

Это и представляет собой определенную *проблему*, особенно, если необходимо обработать значительные объемы исходных данных за ограниченное время.

## **2. Исследование характеристик источников исходных данных**

### **2.1. Источники информации**

Существуют различные источники информации об объектах управления:

– результаты социологических опросов и психологического тестирования;

– выходные файлы информационно-измерительных систем;

– результаты экономического мониторинга, как правило, представленные в форме таблиц Word и реже Excel.

Соответственно, существуют всего три основные формы исходных данных:

1. Рукописные материалы.
2. Файлы стандарта "Текст-DOS".
3. Таблицы Word и Excel.

Возможен также вариант использования для подготовки данных специально разработанных систем окружения, например системы "Эйдос-фонд" [3] или "Эйдос-Ψ" [4] и программных интерфейсов.

В данной статье рассмотрим второй и третий варианты представления исходных данных, т.к. первый вариант, как и вариант с системами окружения, подробно описаны в работе [2].

Примеры небольших фрагментов исходных данных в первой и второй формах представления приведены в таблицах 1 и 2.

**Таблица 1. ПРИМЕР ИСХОДНЫХ ДАННЫХ В СТАНДАРТЕ  
"ТЕКСТ-DOS"**

Код инф. ист.	Наименования полей							
	классы				факторы			
	Obj_1	Obj_2	Obj_3	Obj_4	Atr_1	Atr_2	Atr_3	Atr_4
3028N	70	70	70	70	15	100	124	96
3489N	70	70	70	70	24	-3	35	174
4010N	70	70	70	70	-1	-30	-24	68
4512N	70	70	70	70	-11	19	-35	31
4993N	70	70	70	70	53	-135	48	108
5452N	70	70	70	130	18	-21	139	80
5954N	70	130	70	70	70	84	29	64
6494N	70	70	70	70	-40	-102	-3	65
6955N	70	70	70	70	68	-46	125	100
7476N	70	70	70	70	-21	-19	-51	50

**Таблица 2. ПРИМЕР ИСХОДНЫХ ДАННЫХ  
В СТАНДАРТАХ "ТАБЛИЦА WORD", "ТАБЛИЦА EXCEL"**

№	Наименование показателя и единиц измерения	1998 г.	1999 г.	2000 г.	2001 г.	2002 г.
<b>Зерновые и зернобобовые</b>						
1	Валовой сбор, тыс. т.	3532	6357	6792	7982	8481
2	Уборочная площадь, тыс. га	1464	1884	1967	2103	2045
3	Урожайность, ц/га	24	34	35	39	42
<b>в т.ч. пшеница</b>						
4	Валовой сбор, тыс. т.	1962	4230	4074	5291	5393
5	Уборочная площадь, тыс. га	692	1129	1061	1211	1151
6	Урожайность, ц/га	28	38	38	44	47
<b>Ячмень</b>						
7	Валовой сбор, тыс. т.	806	1214	1471	1811	1825
8	Уборочная площадь, тыс. га	323	336	422	477	464
9	Урожайность, ц/га	25	36	35	39	39
<b>кукуруза на зерно</b>						
10	Валовой сбор, тыс. т.	329	423	587	252	633
11	Уборочная площадь, тыс. га	256	211	267	210	220
12	Урожайность, ц/га	13	20	22	14	30
<b>Рис</b>						
13	Валовой сбор, тыс. т.	314	336	462	393	407
14	Уборочная площадь, тыс. га	92	113	111	99	103
15	Урожайность, ц/га	34	30	42	42	42
<b>Сахарная свекла</b>						
16	Валовой сбор, тыс. т.	2134	2936	2827	3048	4202
17	Уборочная площадь, тыс. га	151	163	126	128	142
18	Урожайность, ц/га	142	180	224	241	295
<b>Подсолнечник</b>						
19	Валовой сбор, тыс. т.	570	613	622	469	732
20	Уборочная площадь, тыс. га	458	472	399	352	424
21	Урожайность, ц/га	12	13	16	14	17
<b>Соя</b>						
22	Валовой сбор, тыс. т.	42	48	54	36	97
23	Уборочная площадь, тыс. га	78	50	48	44	59
24	Урожайность, ц/га	5	10	11	9	18

## **2.2. Характеристики источников информации**

Опыт исследований во многих предметных областях в течение ряда лет показывает, что в исходных данных встречаются показатели различной природы, представляемые в различных единицах измерения. Кроме того, данные часто оказываются неполны (фрагментарны) и зашумлены. Причем восполнить отсутствующие данные обычно нет возможности. Поэтому *возникает проблема выбора метода совместной сопоставимой обработки разнородных по своей природе фрагментарных данных.*

## **2.3. Требования к математической модели**

Анализ характеристик исходных данных показывает, что математическая модель предметной области должна обеспечивать:

- непараметрический анализ разнородных по своей природе неполных (фрагментированных) и зашумленных данных большой размерности;
- выявление и исследование в сопоставимой форме причинно-следственных взаимосвязей факторов среды и управления различной природы, с одной стороны, и будущих состояний объекта управления – с другой.

## **3. Традиционные пути решения проблемы**

Как правило, разработка классификационных и описательных шкал и градаций, их использование для кодирования обучающей выборки, их ввод в обрабатывающую систему – все эти работы выполняются вручную. Однако в ряде случаев вручную кодировать и вводить исходные данные не представляется возможным по причинам:

- 1) жестких ограничений на время, отводящееся заказчиком на решение задачи;
- 2) большой трудоемкости формализации (ручного кодирования) и ввода информации в программную систему;
- 3) большого количества ошибок ручного кодирования и ввода.

В результате решаются только те задачи, для которых допустимое время решения достаточно велико и объемы исходных данных малы, а ошибки ввода малосущественны. При этом большое количество важных для науки и практики задач, которые не вписываются в эти параметры, просто не решаются.

#### **4. Предлагаемое решение проблемы**

Прежде чем осуществить синтез модели в соответствии с методологией системно-когнитивного анализа (СК-анализ), необходимо выполнить первый этап [1], состоящий из следующих шагов:

1. Разработка когнитивной модели предметной области и принятие решения о том, что в модели будет выступать в качестве причин (факторов), а что в качестве следствий (классов).

2. Конструирование классификационных и описательных шкал и градаций.

3. Описание исходной информации в системе шкал и градаций и формирование обучающей выборки.

4. Преобразование исходных данных из формы, в которой они имеются, в стандартную форму, предусмотренную универсальной когнитивной аналитической системой "Эйдос" (далее – система "Эйдос"), представляющую собой инструментарий СК-анализа [5, 6].

1-й этап "Когнитивная структуризация предметной области" является познанием и формированием обобщенной понятийной модели.

2-й, 3-й и 4-й этапы образуют в совокупности этап "Формализация предметной области", т.е. конкретизацию когнитивной модели и преобразование ее в форму, непосредственно пригодную для обработки в программной системе.

*Предлагается решение обозначенной в статье проблемы путем максимальной автоматизации процессов кодирования и ввода исходной*

*информации* в систему "Эйдос", т.е. разработки и использования соответствующего программного интерфейса, работающего с конкретным представлением исходных данных.

В настоящее время этап когнитивной структуризации является в принципе не формализуемым, т.е. может выполняться только человеком. Когда модель уже создана, она позволяет провести системно-когнитивный анализ объекта управления и подтвердить или опровергнуть адекватность решений, принятых при когнитивной структуризации.

Необходимо отметить, что наличие в распоряжении исследователей *инструментария* АСК-анализа и программного интерфейса, автоматизирующего ввод исходной информации и информации обратной связи в систему "Эйдос", позволяет не только осуществить синтез СИМ, но и *периодически осуществлять адаптацию и синтез новых версий семантической информационной модели*. Таким образом обеспечивается корректировка решений, принятых на первых этапах синтеза модели, а также отслеживание динамики предметной области, благодаря чему сохраняется высокая адекватность модели в изменяющихся условиях.

Шагам первого этапа СК-анализа соответствуют четыре функции, поддерживаемые программным интерфейсом:

1. Определение классов и факторов.
2. Выявление макропараметров, определяющих состояния объекта управления (классы); измерение области изменения числовых значений факторов и конструирование интервалов (диапазонов) факторов; конструирование классификационных и описательных шкал и градаций и их кодирование.
3. Кодирование исходных данных в системе классификационных и описательных шкал и градаций и формирование обучающей выборки (базы прецедентов).

4. Автоматический ввод классификационных и описательных шкал и градаций, а также обучающей выборки в соответствующие стандартные базы данных системы "Эйдос".

Реально в программном интерфейсе, реализованном на языке программирования, поддерживаются все 4 функции, а в Excel – только первые 3.

## **5. Когнитивная структуризация предметной области**

### **5.1. Задачи когнитивной структуризации**

Под когнитивной структуризацией предметной области понимается процесс ее познания, который осуществляется на основе *системного подхода*, в соответствии с которым объект познания рассматривается как *система*, имеющая сложное многоуровневое иерархическое строение. Когнитивная структуризация предметной области – это начальный этап синтеза модели, подготавливающий формализацию и предшествующий ей.

При когнитивной структуризации:

- выделяются *целевые параметры системы*, т.е. ее желательные и нежелательные *будущие состояния*, характеризующие ее на макроуровне;
- определяется *система факторов*, детерминирующих эти будущие состояния.

При этом в качестве факторов могут рассматриваться (окружающая среда, технологии, а также параметры системы на низких уровнях ее иерархической структуры).

### **5.2. Будущие состояния объекта управления (классы)**

В качестве будущих состояний объекта управления выбираются:

- в варианте "Текст-DOS" – интервалы макропараметров в столбцах от Obj\_1 до Obj\_W;
- в варианте Excel – интервалы макропараметров в строках, определенных при когнитивной структуризации.

### 5.3. Факторы

В качестве будущих факторов выбираются:

- в варианте "Текст-DOS" – интервалы значений параметров в столбцах от Atr\_1 до Atr\_M;
- в варианте Excel – интервалы параметров в строках, определенных при когнитивной структуризации.

## 6. Формализация предметной области

### 6.1. Общие положения

Формализация предметной области осуществляется на основе ее предварительной когнитивной структуризации. Формализация предметной области – это конструирование классификационных и описательных шкал и градаций, как правило, порядкового типа с применением интервальных оценок, в системе которых предметная область описывается в форме, пригодной для обработки на компьютере с использованием математических моделей.

В соответствии с методом СК-анализа, каждый числовой фактор из таблиц 1 или 2, независимо от его смысла и единиц измерения, рассматривается как переменная числовая величина, принимающая определенное множество значений. Подобные величины формализуются путем сведения к *интервальным значениям*, т.е.

- введения некоторого количества диапазонов, охватывающих все множество значений фактора;
- установления фактов попадания конкретных значений величин в определенные диапазоны.

Для каждого фактора устанавливаются свои *границы диапазонов*, исходя из их количества и *множества значений* величины фактора.

*Количество* диапазонов может быть различным для разных факторов, но на практике удобнее выбирать его одинаковым для всех фактов.

Рассмотрим основные соображения, из которых исходят при выборе количества диапазонов. Чем больше диапазонов, тем точнее интервальные оценки. Однако это верно только тогда, когда, по крайней мере, для большинства диапазонов наблюдаются факты попадания значений факторов в них. Очевидно, для этого необходимо достаточно большое количество данных. Если их недостаточно, то многие диапазоны могут оказаться пустыми и модель приближается к детерминистскому типу. В этом случае имеет смысл уменьшить их количество, укрупнить их. При определении количества диапазонов необходимо учитывать также возможные ограничения базовой программной системы, в данном случае системы "Эйдос".

Из этих рассуждений следует вывод о том, что при большом количестве данных оправданно увеличить количество диапазонов и повысить точность исследования. Когда же данных недостаточно, приходится укрупнять диапазоны, что приводит к некоторой вынужденной неточности выводов, но делает их более обоснованными статистически. По-видимому, это утверждение можно считать одним из следствий теоремы Котельникова об отсчетах.

СК-анализ предусматривает также возможность использования вторичных показателей, являющихся различными функциями первичных показателей. Однако этот подход требует данных большого объема за достаточно длительный период времени.

## **6.2. Программный интерфейс формализации предметной области при исходных данных типа "Текст-DOS"**

Ниже приводится полный исходный текст реального программного интерфейса на языке программирования xBASE, реализующего все перечисленные выше 4 функции в соответствии с алгоритмом, приведенным выше.

```

***** ФОРМИРОВАНИЕ КЛАССИФИКАЦИОННЫХ И ОПИСАТЕЛЬНЫХ ШКАЛ И ГРАДАЦИЙ *****
***** И ОБУЧАЮЩЕЙ ВЫБОРКИ ИЗ СТАНДАРТНОГО DBF-Excel-файла *****
***** ПРОФ.АРТУРА НИКОЛАЕВИЧА ЛЕБЕДЕВА И ТИМУРА ЩУКИНА *****
***** Луценко Е.В., 02/05/04 09:02am *****

```

```
PARAMETERS File_name
```

```
***** БЛОК-1. ОТОБРАЖЕНИЕ ИНФОРМАЦИИ О ФУНКЦИЯХ ПРОГРАММНОГО ИНТЕРФЕЙСА *****
```

```
scr23 = SAVESCREEN(0,0,24,79)
```

```
SET CURSOR OFF
SET DATE ITALIAN
SET DECIMALS TO 15
SET ESCAPE On
```

```
FOR J=0 TO 24
```

```
    @J,0 SAY REPLICATE(" ",80) COLOR "rg+N"
```

```
NEXT
```

```
SHOWTIME(0,60,.F.,"rg+n",.F.,.F.)
```

```
Mess1 = " === ГЕНЕРАЦИЯ КЛАССИФИКАЦИОННЫХ ШКАЛ И ГРАДАЦИЙ === "
```

```
Mess2 = " === ГЕНЕРАЦИЯ ОПИСАТЕЛЬНЫХ ШКАЛ И ГРАДАЦИЙ === "
```

```
Mess3 = " === ГЕНЕРАЦИЯ ОБУЧАЮЩЕЙ ВЫБОРКИ === "
```

```
@1,40-LEN(Mess1)/2 SAY Mess1 COLOR "rg+rb"
```

```
@2,40-LEN(Mess2)/2 SAY Mess2 COLOR "rg+rb"
```

```
@3,40-LEN(Mess3)/2 SAY Mess3 COLOR "rg+rb"
```

```
** БЛОК-2. ПРОВЕРКА КОРРЕКТНОСТИ ИМЕНИ ФАЙЛА С ИСХОДНЫМИ ДАННЫМИ *****
```

```
Fn = File_name
```

```
IF EMPTY(Fn)
```

```
    Mess = "Программа запускается с параметром: INPTIMUR.EXE <File_name>"
```

```
    @15,40-LEN(Mess)/2 SAY Mess COLOR "rg+rb"
```

```
    INKEY(0)
```

```
    RESTSCREEN(0,0,24,79,scr23)
```

```
    CLOSE ALL
```

```
    QUIT
```

```
ENDIF
```

```
Pos_p = AT(".",Fn)
```

```
IF Pos_p = 0
```

```
    Mess = "У файла, указанного в качестве параметра, должен быть тип DBF"
```

```
    @15,40-LEN(Mess)/2 SAY Mess COLOR "rg+rb"
```

```
    INKEY(0)
```

```
    RESTSCREEN(0,0,24,79,scr23)
```

```
    CLOSE ALL
```

```
    QUIT
```

```
ENDIF
```

```
Ext = UPPER(SUBSTR(Fn,Pos_p+1))
```

```
IF Ext <> "DBF"
```

```

Mess = "У файла, указанного в качестве параметра, должно быть расширение DBF"
@15,40-LEN(Mess)/2 SAY Mess COLOR "rg+/rb"

INKEY(0)
RESTSCREEN(0,0,24,79,scr23)
CLOSE ALL
QUIT
ENDIF

Fns = ALLTRIM(SUBSTR(File_name,1,Pos_p-1))

IF LEN(Fns) > 5
  Mess = "Длина имени файла, указанного параметром, должна быть <= 5 символов"
  @15,40-LEN(Mess)/2 SAY Mess COLOR "rg+/rb"
  INKEY(0)
  RESTSCREEN(0,0,24,79,scr23)
  CLOSE ALL
  QUIT
ENDIF

** БЛОК-3. ЗАДАНИЕ В ДИАЛОГЕ ДИАПАЗОНОВ СТОЛБЦОВ С КЛАССАМИ И ФАКТОРАМИ *****

M_KodObj1 = 2
M_KodObj2 = 15
Mess = "Задайте диапазон номеров столбцов классов : #### ####"
@8,40-LEN(Mess)/2 SAY Mess COLOR "w+/rb"
@8,58 GET M_KodObj1 PICTURE "####" COLOR "rg+/r+"
@8,63 GET M_KodObj2 PICTURE "####" COLOR "rg+/r+"

M_KodAtr1 = 16
M_KodAtr2 = 49
Mess = "Задайте диапазон номеров столбцов признаков: #### ####"
@10,40-LEN(Mess)/2 SAY Mess COLOR "w+/rb"
@10,58 GET M_KodAtr1 PICTURE "####" COLOR "rg+/r+"
@10,63 GET M_KodAtr2 PICTURE "####" COLOR "rg+/r+"

SET CURSOR ON;READ;SET CURSOR OFF
IF LASTKEY()=27 .OR. M_KodObj1>M_KodObj2 .OR. M_KodAtr1>M_KodAtr2
  RESTSCREEN(0,0,24,79,scr23)
  RETURN 2
ENDIF

** БЛОК-4. РАСЧЕТ И ВЫВОД ИНФОРМАЦИИ О КОЛИЧЕСТВЕ КЛАССИФИКАЦИОННЫХ И ОПИСАТЕЛЬНЫХ ШКАЛ *****

USE Object EXCLUSIVE NEW;ZAP
USE (Fns) EXCLUSIVE NEW

N_OpSh = M_KodAtr2 - M_KodAtr1 + 1      && Кол-во описательных шкал
N_KlSh = M_KodObj2 - M_KodObj1 + 1      && Кол-во классификационных шкал

N_Rec   = RECCOUNT()
M_Kod   = 0

```

```

Mess = "КОЛИЧЕСТВО КЛАССИФИКАЦИОННЫХ ШКАЛ: "+ALLTRIM(STR(N_KlSh,5))
@14,40-LEN(Mess)/2 SAY Mess COLOR "w+/rb"
Mess = "КОЛИЧЕСТВО ОПИСАТЕЛЬНЫХ ШКАЛ: "+ALLTRIM(STR(N_OpSh,5))
@15,40-LEN(Mess)/2 SAY Mess COLOR "w+/rb"

** БЛОК-5. РАСЧЕТ И ВЫВОД ИНФОРМАЦИИ О ПРЕДЕЛЬНОМ КОЛИЧЕСТВЕ ГРАДАЦИЙ ОПИСАТЕЛЬНЫХ ШКАЛ *****

PUBLIC Nmax_Grad := INT(4000/(N_OpSh))

Mess = "Градаций в описательной шкале не более: "+ALLTRIM(STR(Nmax_Grad,5))
@18,40-LEN(Mess)/2 SAY Mess COLOR "w+/rb"

** БЛОК-6. ЗАДАНИЕ В ДИАЛОГЕ КОЛИЧЕСТВА ГРАДАЦИЙ (ИНТЕРВАЛОВ) В ОПИСАТЕЛЬНЫХ ШКАЛАХ *****

Mess = "Задайте максимальное количество градаций в описательных шкалах: ####"
@20,40-LEN(Mess)/2 SAY Mess COLOR "w+/rb"
N_Grad = Nmax_Grad
@20,70 GET N_Grad PICTURE "####" COLOR "rg+/r"

SET CURSOR ON;READ;SET CURSOR OFF
IF LASTKEY()=27 .OR. N_Grad > Nmax_Grad
    RESTSCREEN(0,0,24,79,scr23)
    RETURN 2
ENDIF

** БЛОК-7. ГЕНЕРАЦИЯ КЛАССИФИКАЦИОННЫХ ШКАЛ И ГРАДАЦИЙ *****

Mess1 = " === ГЕНЕРАЦИЯ КЛАССИФИКАЦИОННЫХ ШКАЛ И ГРАДАЦИЙ === "
@1,40-LEN(Mess1)/2 SAY Mess1 COLOR "rg+*/rb"

FOR f=M_KodObj1 TO M_KodObj2          && Цикл по полям классов БД &Fns
    SELECT &Fns
    Fn = ALLTRIM(FIELDNAME(f))        && Принадлежность к профессиональному типу
    INDEX ON STR(999999-&Fn,7) TO Mrk_uniq UNIQUE
    DBGOTOP()
    DO WHILE .NOT. EOF()
        Fv = ALLTRIM(STR(FIELDGET(f)))
        SELECT Object
        APPEND BLANK
        REPLACE Kod WITH ++M_Kod
        REPLACE Name WITH "MRK-"+Fn+"-"+Fv
        SELECT &Fns
        DBSKIP(1)
    ENDDO
NEXT

CLOSE ALL

** БЛОК-8. ГЕНЕРАЦИЯ ОПИСАТЕЛЬНЫХ ШКАЛ И ГРАДАЦИЙ *****

```

```

Mess2 = " === ГЕНЕРАЦИЯ ОПИСАТЕЛЬНЫХ ШКАЛ И ГРАДАЦИЙ === "
@2,40-LEN(Mess2)/2 SAY Mess2 COLOR "rg+*/rb"

Fns_sh = Fns   + "_sh"      && БД для границ интервалов для SELECT
Fnd     = Fns   + ".dbf"    && БД для границ интервалов для COPY
Fnd_sh  = Fns_sh + ".dbf"   && БД для границ интервалов для COPY

COPY FILE (Fnd) TO (Fnd_sh)

CLOSE ALL

USE Priz_ob EXCLUSIVE NEW;ZAP
USE Priz_per EXCLUSIVE NEW;ZAP
USE (Fns)   EXCLUSIVE NEW
USE (Fns_sh) EXCLUSIVE NEW;ZAP;APPEND BLANK;APPEND BLANK;APPEND BLANK

SELECT &Fns
N_Rec  = RECCOUNT()
Kod_prob= 0
Kod_prp = 0

@24,0 SAY REPLICATE("-",80) COLOR "rb/n"

FOR ff=M_KodAtr1 TO M_KodAtr2      && Начало цикла по полям БД &Fns

    SELECT &Fns
    F_Name = FIELDNAME(ff)
    INDEX ON STR(999999-FIELDGET(ff),7) TO Mrk_funi UNIQUE
    DBGOTOP(); F_MaxSH = FIELDGET(ff)
    DBGOBOTTOM();F_MinSH = FIELDGET(ff)
    Delta = (F_MaxSH-F_MinSH)/N_Grad

    SELECT &Fns_sh
    DBGOTO(1);FIELDPUT(ff,F_MaxSH)
    DBGOTO(2);FIELDPUT(ff,F_MinSH)
    DBGOTO(3);FIELDPUT(ff,Delta)

    SELECT Priz_ob
    APPEND BLANK
    REPLACE Kod WITH ++Kod_prob
    REPLACE Name WITH F_Name

@23,0 SAY REPLICATE("-",80) COLOR "rb/n"

FOR gr=1 TO N_Grad

    SELECT Priz_per
    APPEND BLANK
    F_MinGR = F_MinSH+(gr-1)*Delta
    F_MaxGR = F_MinSH+(gr )*Delta
    M_Name = F_Name+": {"+ALLTRIM(STR(F_MinGR,5))+", "+ALLTRIM(STR(F_MaxGR,5))+"}"
    REPLACE Kod WITH ++Kod_prp

```

```

REPLACE Kod_ob_pr WITH Kod_prob
REPLACE Name WITH M_Name

SELECT Priz_ob
FIELDPUT(gr+2,Kod_prp)

p=gr/N_Grad*100;p=IF(p<=100,p,100)
@23,0 SAY STR(p,3)+"%" COLOR "w+/r+"
@23,4 SAY REPLICATE("-",0.76*p) COLOR "rb+/n"
NEXT

p=ff/N_OpSh*100;p=IF(p<=100,p,100)
@24,0 SAY STR(p,3)+"%" COLOR "w+/r+"
@24,4 SAY REPLICATE("-",0.76*p) COLOR "rb+/n"

NEXT

CLOSE ALL

** БЛОК-9. ГЕНЕРАЦИЯ ОБУЧАЮЩЕЙ ВЫБОРКИ *****

Mess3 = " === ГЕНЕРАЦИЯ ОБУЧАЮЩЕЙ ВЫБОРКИ === "
@3,40-LEN(Mess3)/2 SAY Mess3 COLOR "rg+*/rb"

USE Object EXCLUSIVE NEW
INDEX ON Name TO Obj_name

USE Priz_per EXCLUSIVE NEW
INDEX ON Name TO Prpe_nam

CLOSE ALL

USE Object INDEX Obj_name EXCLUSIVE NEW
USE Priz_per INDEX Prpe_nam EXCLUSIVE NEW
USE (Fns) EXCLUSIVE NEW
USE (Fnd_sh) EXCLUSIVE NEW
USE ObInfZag EXCLUSIVE NEW;ZAP
USE ObInfKpr EXCLUSIVE NEW;ZAP

N_Rec = RECCOUNT()
DBGOTOP()

@24,0 SAY REPLICATE("-",80) COLOR "rb/n"

M_KodIst = 0

SELECT &Fns
N_Rec = RECCOUNT()
Num = 0

DBGOTOP()

```

```

DO WHILE .NOT. EOF()

A_Obj = {}
***** База заголовков
SELECT &Fns
M_NameIst = FIELDGET(1)

FOR f=M_KodObj1 TO M_KodObj2      && Цикл по полям классов БД &Fns
  SELECT &Fns
  Fn = ALLTRIM(FIELDNAME(f))      && Принадлежность к профессиональному типу
  Fv = ALLTRIM(STR(FIELDGET(f)))
  SELECT Object;SET ORDER TO 1;T=DBSEEK("MRK-"+Fn+"-"+Fv)
  IF T
    AADD(A_Obj, Kod)
  ENDIF
NEXT
ASORT(A_Obj)

SELECT ObInfZag
APPEND BLANK
REPLACE Kod_ist WITH ++M_KodIst
REPLACE Name_ist WITH M_NameIst
FOR jj=1 TO LEN(A_Obj)
  FIELDPUT(jj+2,A_Obj[jj])
NEXT

***** База признаков
SELECT &Fns
***** Формирование массива первичных признаков из БД &Fns
Ar = {}
FOR ff=M_KodAtr1 TO M_KodAtr2      && Начало цикла по полям БД &Fns
  AADD(Ar, FIELDGET(ff))
NEXT

***** Формирование массива кодов признаков из БД &Fns
M_KodPr = {}
FOR jj=1 TO LEN(Ar)
  SELECT &Fns_sh
  F_Name = FIELDNAME(jj+M_KodAtr1-1)
  DBGOTO(1);F_MaxSH = FIELDGET(jj+M_KodAtr1-1)
  DBGOTO(2);F_MinSH = FIELDGET(jj+M_KodAtr1-1)
  Delta = (F_MaxSH-F_MinSH)/N_Grad
  FOR gr=1 TO N_Grad
    F_MinGR = F_MinSH+(gr-1)*Delta
    F_MaxGR = F_MinSH+(gr )*Delta
    IF F_MinGR <= Ar[jj] .AND. Ar[jj] <= F_MaxGR
      M_Name = F_Name+": {"+ALLTRIM(STR(F_MinGR,5))+", "+ALLTRIM(STR(F_MaxGR,5))+"}"
      SELECT Priz_per;SET ORDER TO 1;T=DBSEEK(M_Name)
      IF T
        AADD(M_KodPr, Kod)
      ENDIF
    ENDIF
  ENDIF
NEXT

```

```

ENDIF
NEXT
NEXT

***** Запись массива кодов признаков из БД &Fns в БД ObInfKpr
SELECT ObInfKpr
APPEND BLANK
FIELDPUT(1,M_KodIst)
k=2
FOR jj=1 TO LEN(M_KodPr)
  IF k <= 12
    FIELDPUT(k++,M_KodPr[jj])
  ELSE
    APPEND BLANK
    FIELDPUT(1,M_KodIst)
    k=2
    FIELDPUT(k ,M_KodPr[jj])
  ENDIF
NEXT
p=++Num/N_Rec*100;p=IF(p<=100,p,100)
@24,0 SAY STR(p,3)+"%" COLOR "w+/r+"
@24,4 SAY REPLICATE("-",0.76*p) COLOR "rg+/n"
SELECT &Fns
DBSKIP(1)
ENDDO

Mess = " ПРОЦЕСС ГЕНЕРАЦИИ ЗАВЕРШЕН УСПЕШНО !!! "
@24,40-LEN(Mess)/2 SAY Mess COLOR "rg+/rb"

INKEY(0)

RESTSCREEN(0,0,24,79,scr23)
CLOSE ALL

QUIT

```

Этот программный интерфейс работает с исходными данными, получающимися из данных типа "Текст-DOS" после загрузки их в Excel и записи в DBF-файл (тип файла DBF DBase IV). Загрузка файла "Текст-DOS" осуществляется по технологии, представленной в работе [7]. При запуске программы DBF-файл с исходными данными указывается в качестве параметра. В диалоге задаются номера столбцов с информацией по классам и факторам, а также количество интервалов (градаций в описательных шкалах). Количество объектов обучающей выборки неограниченно.

В результате работы программного интерфейса *автоматически* формируются справочники классификационных и описательных шкал и градаций, а также обучающая выборка (табл. 3 и 4), что позволяет непосредственно приступить к выполнению последующих этапов СК-анализа, т.е. к синтезу модели, ее оптимизации и проверке на адекватность и т.д., которые осуществляются средствами самой системы "Эйдос".

**Таблица 3. КЛАССИФИКАЦИОННЫЕ И ОПИСАТЕЛЬНЫЕ ШКАЛЫ И ГРАДАЦИИ В СТАНДАРТЕ "ТЕКСТ-DOS"**

КЛАССИФИКАЦИОННЫЕ ШКАЛЫ И ГРАДАЦИИ		ОПИСАТЕЛЬНЫЕ ШКАЛЫ И ГРАДАЦИИ	
Код	Наименование	Код	Наименование
1	MRK-T01-70..	[ 1]  AVM	[ 3]  M02
2	MRK-T02-130..	1  AVM: {-40, -18}.	11  M02: {-51, -13}.
3	MRK-T02-70..	2  AVM: {-18, 4}...	12  M02: {-13, 25}..
4	MRK-T03-70..	3  AVM: {4, 26}....	13  M02: {25, 63}...
5	MRK-T04-130..	4  AVM: {26, 48}...	14  M02: {63, 101}..
6	MRK-T04-70..	5  AVM: {48, 70}...	15  M02: {101, 139}.
		[ 2]  M01	[ 4]  M03
		6  M01: {-135, -88}	16  M03: {31, 60}...
		7  M01: {-88, -41}.	17  M03: {60, 88}...
		8  M01: {-41, 6}...	18  M03: {88, 117}..
		9  M01: {6, 53}....	19  M03: {117, 145}.
		10  M01: {53, 100}..	20  M03: {145, 174}.

**Таблица 4. ОБУЧАЮЩАЯ ВЫБОРКА В СТАНДАРТЕ "ТЕКСТ-DOS"**

Объект обуч. выб.	Коды классов	Коды факторов
-------------------	--------------	---------------

Код	Наим.	Obj_1	Obj_2	Obj_3	Obj_4	Atr_1	Atr_2	Atr_3	Atr_4
1	3028N	1	3	4	6	3	10	15	18
2	3489N	1	3	4	6	3	8	13	20
3	4010N	1	3	4	6	2	8	11	17
4	4512N	1	3	4	6	2	9	11	16
5	4993N	1	3	4	6	5	6	13	18
6	5452N	1	3	4	5	3	8	15	17
7	5954N	1	2	4	6	5	10	13	17
8	6494N	1	3	4	6	1	6	12	17
9	6955N	1	3	4	6	5	7	15	18
10	7476N	1	3	4	6	1	8	11	16

### 6.3. Инструментарий работы с исходными данными типа "Документ Word" или "Таблица Excel"

Прежде всего, таблицы из документов Word преобразуются в таблицы Excel следующим образом:

1. Выделяем блоком таблицу Word.
2. Помещаем блок в буфер обмена.
3. Вставляем таблицу из буфера обмена в Excel.
4. Форматируем таблицу в Excel (оформляем заголовки и шапки, задаем форматы вывода текстовых и числовых значений, устанавливаем высоту строки и ширину столбцов, создаем сетку разметки и т.п.).
5. Определяем области значений классов и факторов, т.е. минимальное и максимальное значения каждого класса и фактора.
6. Получаем величину интервалов для каждого класса и фактора исходя из их областей значений и количества интервалов.
7. Находим минимальное и максимальное значения для каждого интервала.

В результате выполнения этих шагов из исходных данных, представленных в форме таблицы 2, получаем таблицу 5

**Таблица 5. ОБЛАСТИ ЗНАЧЕНИЙ И ГРАНИЦЫ ДИАПАЗОНОВ  
ИЗМЕНЕНИЯ ФАКТОРОВ В СТАНДАРТАХ "ТАБЛИЦА WORD",  
"ТАБЛИЦА EXCEL"**

№	Наименование показателя	Область значений		D	Диапазоны (интервалы)									
		Min	Max		1-й		2-й		3-й		4-й		5-й	
					Min	Max	Min	Max	Min	Max	Min	Max	Min	Max
	<b>Зерновые и зерно-бобовые</b> <small>И ед. измерения</small>													
1	Валовой сбор, тыс. т.	3532	8481	990	3532	4522	4522	5512	5512	6502	6502	7491	7491	8481
2	Уборочная площадь, тыс. га	1464	2103	128	1464	1592	1592	1719	1719	1847	1847	1975	1975	2103
3	Урожайность, ц/га	24	42	3	24	28	28	31	31	35	35	38	38	42
	<b>в т.ч. пшеница</b>													
4	Валовой сбор, тыс. т.	1962	5393	686	1962	2648	2648	3334	3334	4021	4021	4707	4707	5393
5	Уборочная площадь, тыс. га	692	1211	104	692	796	796	900	900	1004	1004	1107	1107	1211
6	Урожайность, ц/га	28	47	4	28	32	32	36	36	39	39	43	43	47
	<b>Ячмень</b>													
7	Валовой сбор, тыс. т.	806	1825	204	806	1010	1010	1214	1214	1417	1417	1621	1621	1825
8	Уборочная площадь, тыс. га	323	477	31	323	354	354	385	385	415	415	446	446	477
9	Урожайность, ц/га	25	39	3	25	28	28	31	31	34	34	36	36	39
	<b>кукуруза на зерно</b>													
10	Валовой сбор, тыс. т.	252	633	76	252	328	328	404	404	481	481	557	557	633
11	Уборочная площадь, тыс. га	210	267	11	210	221	221	233	233	244	244	255	255	267
12	Урожайность, ц/га	13	30	3	13	16	16	20	20	23	23	26	26	30
	<b>Рис</b>													
13	Валовой сбор, тыс. т.	314	462	30	314	344	344	373	373	403	403	432	432	462

	т.													
14	Уборочная площадь, тыс. га	92	113	4	92	96	96	100	100	104	104	109	109	113
15	Урожайность, ц/га	30	42	3	30	32	32	35	35	37	37	40	40	42
<b>Сахарная свекла</b>														
16	Валовой сбор, тыс. т.	2134	4202	414	2134	2548	2548	2961	2961	3375	3375	3789	3789	4202
17	Уборочная площадь, тыс. га	126	163	7	126	134	134	141	141	148	148	156	156	163
18	Урожайность, ц/га	142	295	31	142	173	173	203	203	234	234	264	264	295
<b>Подсолнечник</b>														
19	Валовой сбор, тыс. т.	469	732	53	469	521	521	574	574	627	627	680	680	732
20	Уборочная площадь, тыс. га	352	472	24	352	376	376	400	400	424	424	448	448	472
21	Урожайность, ц/га	12	17	1	12	13	13	14	14	15	15	16	16	17
<b>Соя</b>														
22	Валовой сбор, тыс. т.	36	97	12	36	48	48	60	60	72	72	84	84	97
23	Уборочная площадь, тыс. га	44	78	7	44	51	51	58	58	65	65	71	71	78
24	Урожайность, ц/га	5	18	2	5	8	8	10	10	13	13	15	15	18

Используя данные таблицы 5, сконструируем классификационные шкалы и градации (табл. 6 и 7)

**Таблица 6. КЛАССИФИКАЦИОННЫЕ ШКАЛЫ И ГРАДАЦИИ  
В СТАНДАРТАХ "ТАБЛИЦА WORD", "ТАБЛИЦА EXCEL"**

Код	Наименования классификационных шкал и градаций
1	ВАЛОВОЙ СБОР - очень низкий
2	ВАЛОВОЙ СБОР – низкий
3	ВАЛОВОЙ СБОР - средний
4	ВАЛОВОЙ СБОР - высокий
5	ВАЛОВОЙ СБОР - очень высокий
6	ВАЛОВОЙ СБОР ЗЕРНОВЫХ И ЗЕРНОБОБОВЫХ - очень низкий
7	ВАЛОВОЙ СБОР ЗЕРНОВЫХ И ЗЕРНОБОБОВЫХ - низкий

8	ВАЛОВОЙ СБОР ЗЕРНОВЫХ И ЗЕРНОБОБОВЫХ - средний
9	ВАЛОВОЙ СБОР ЗЕРНОВЫХ И ЗЕРНОБОБОВЫХ - высокий
10	ВАЛОВОЙ СБОР ЗЕРНОВЫХ И ЗЕРНОБОБОВЫХ - очень высокий
11	ВАЛОВОЙ СБОР ПШЕНИЦЫ - очень низкий
12	ВАЛОВОЙ СБОР ПШЕНИЦЫ - низкий
13	ВАЛОВОЙ СБОР ПШЕНИЦЫ - средний
14	ВАЛОВОЙ СБОР ПШЕНИЦЫ - высокий
15	ВАЛОВОЙ СБОР ПШЕНИЦЫ - очень высокий
16	ВАЛОВОЙ СБОР ЯЧМЕНЯ - очень низкий
17	ВАЛОВОЙ СБОР ЯЧМЕНЯ - низкий
18	ВАЛОВОЙ СБОР ЯЧМЕНЯ - средний
19	ВАЛОВОЙ СБОР ЯЧМЕНЯ - высокий
20	ВАЛОВОЙ СБОР ЯЧМЕНЯ - очень высокий
21	ВАЛОВОЙ СБОР КУКУРУЗЫ - очень низкий
22	ВАЛОВОЙ СБОР КУКУРУЗЫ - низкий
23	ВАЛОВОЙ СБОР КУКУРУЗЫ - средний
24	ВАЛОВОЙ СБОР КУКУРУЗЫ - высокий
25	ВАЛОВОЙ СБОР КУКУРУЗЫ - очень высокий
26	ВАЛОВОЙ СБОР РИСА - очень низкий
27	ВАЛОВОЙ СБОР РИСА - низкий
28	ВАЛОВОЙ СБОР РИСА - средний
29	ВАЛОВОЙ СБОР РИСА - высокий
30	ВАЛОВОЙ СБОР РИСА - очень высокий
31	ВАЛОВОЙ СБОР САХАРНОЙ СВЕКЛЫ - очень низкий
32	ВАЛОВОЙ СБОР САХАРНОЙ СВЕКЛЫ - низкий
33	ВАЛОВОЙ СБОР САХАРНОЙ СВЕКЛЫ - средний
34	ВАЛОВОЙ СБОР САХАРНОЙ СВЕКЛЫ - высокий
35	ВАЛОВОЙ СБОР САХАРНОЙ СВЕКЛЫ - очень высокий
36	ВАЛОВОЙ СБОР ПОДСОЛНЕЧНИКА - очень низкий
37	ВАЛОВОЙ СБОР ПОДСОЛНЕЧНИКА - низкий
38	ВАЛОВОЙ СБОР ПОДСОЛНЕЧНИКА - средний
39	ВАЛОВОЙ СБОР ПОДСОЛНЕЧНИКА - высокий
40	ВАЛОВОЙ СБОР ПОДСОЛНЕЧНИКА - очень высокий
41	ВАЛОВОЙ СБОР СОИ - очень низкий
42	ВАЛОВОЙ СБОР СОИ - низкий
43	ВАЛОВОЙ СБОР СОИ - средний
44	ВАЛОВОЙ СБОР СОИ - высокий
45	ВАЛОВОЙ СБОР СОИ - очень высокий

**Таблица 7. ОПИСАТЕЛЬНЫЕ ШКАЛЫ И ГРАДАЦИИ  
В СТАНДАРТАХ "ТАБЛИЦА WORD", "ТАБЛИЦА EXCEL"**

<b>Код</b>	<b>Наименования описательных шкал и градаций</b>	<b>Код</b>	<b>Наименования описательных шкал и градаций</b>
	УБОРОЧНАЯ ПЛОЩАДЬ ЗЕРНОВЫХ И ЗЕРНОБОБОВЫХ:		УРОЖАЙНОСТЬ ЗЕРНОВЫХ И ЗЕРНОБОБОВЫХ:
<b>1</b>	очень низкая	<b>41</b>	очень низкая
<b>2</b>	Низкая	<b>42</b>	низкая
<b>3</b>	Средняя	<b>43</b>	средняя
<b>4</b>	Высокая	<b>44</b>	высокая
<b>5</b>	очень высокая	<b>45</b>	очень высокая
	УБОРОЧНАЯ ПЛОЩАДЬ ПШЕНИЦЫ:		УРОЖАЙНОСТЬ ПШЕНИЦЫ:
<b>6</b>	очень низкая	<b>46</b>	очень низкая
<b>7</b>	Низкая	<b>47</b>	низкая
<b>8</b>	Средняя	<b>48</b>	средняя
<b>9</b>	Высокая	<b>49</b>	высокая
<b>10</b>	очень высокая	<b>50</b>	очень высокая
	УБОРОЧНАЯ ПЛОЩАДЬ ЯЧМЕНЯ:		УРОЖАЙНОСТЬ ЯЧМЕНЯ:
<b>11</b>	очень низкая	<b>51</b>	очень низкая
<b>12</b>	Низкая	<b>52</b>	низкая
<b>13</b>	Средняя	<b>53</b>	средняя
<b>14</b>	Высокая	<b>54</b>	высокая
<b>15</b>	очень высокая	<b>55</b>	очень высокая
	УБОРОЧНАЯ ПЛОЩАДЬ КУКУРУЗЫ:		УРОЖАЙНОСТЬ КУКУРУЗЫ:
<b>16</b>	очень низкая	<b>56</b>	очень низкая
<b>17</b>	Низкая	<b>57</b>	низкая
<b>18</b>	Средняя	<b>58</b>	средняя
<b>19</b>	Высокая	<b>59</b>	высокая
<b>20</b>	очень высокая	<b>60</b>	очень высокая
	УБОРОЧНАЯ ПЛОЩАДЬ РИСА:		УРОЖАЙНОСТЬ РИСА:
<b>21</b>	очень низкая	<b>61</b>	очень низкая
<b>22</b>	Низкая	<b>62</b>	низкая
<b>23</b>	Средняя	<b>63</b>	средняя
<b>24</b>	Высокая	<b>64</b>	высокая
<b>25</b>	очень высокая	<b>65</b>	очень высокая
	УБОРОЧНАЯ ПЛОЩАДЬ		УРОЖАЙНОСТЬ

	САХАРНОЙ СВЕКЛЫ:		САХАРНОЙ СВЕКЛЫ:
<b>26</b>	очень низкая	<b>66</b>	очень низкая
<b>27</b>	Низкая	<b>67</b>	низкая
<b>28</b>	Средняя	<b>68</b>	средняя
<b>29</b>	Высокая	<b>69</b>	высокая
<b>30</b>	очень высокая	<b>70</b>	очень высокая
	УБОРОЧНАЯ ПЛОЩАДЬ ПОДСОЛНЕЧНИКА:		УРОЖАЙНОСТЬ ПОДСОЛНЕЧНИКА:
<b>31</b>	очень низкая	<b>71</b>	очень низкая
<b>32</b>	Низкая	<b>72</b>	низкая
<b>33</b>	Средняя	<b>73</b>	средняя
<b>34</b>	Высокая	<b>74</b>	высокая
<b>35</b>	очень высокая	<b>75</b>	очень высокая
	УБОРОЧНАЯ ПЛОЩАДЬ СОИ:		УРОЖАЙНОСТЬ СОИ:
<b>36</b>	очень низкая	<b>76</b>	очень низкая
<b>37</b>	Низкая	<b>77</b>	низкая
<b>38</b>	Средняя	<b>78</b>	средняя
<b>39</b>	Высокая	<b>79</b>	высокая
<b>40</b>	очень высокая	<b>80</b>	очень высокая

### **7. Подготовка обучающей выборки**

После реализации этапов конструирования классификационных и описательных шкал и градаций выполняется следующий этап: подготовка базы прецедентов (обучающей выборки). Обучающая выборка (табл. 8) представляет собой исходные данные (см. табл. 2), закодированные с использованием классификационных и описательных шкал и градаций (см. табл. 6 и 7), и состоит из двух взаимосвязанных таблиц:

- таблицы кодов классов;
- таблицы кодов градаций факторов.

**Таблица 8. ОБУЧАЮЩАЯ ВЫБОРКА В СТАНДАРТАХ  
"ТАБЛИЦА WORD", "ТАБЛИЦА EXCEL"**

№	Наименование показателя и единиц измерения	1998 г.	1999 г.	2000 г.	2001 г.	2002 г.
<b>Зерновые и зернобобовые</b>						
1	Валовой сбор, тыс. т.	6	8	9	10	10
2	Уборочная площадь, тыс. га	1	4	4	5	5
3	Урожайность, ц/га	41	43	43	45	45
<b>в т.ч. пшеница</b>						
4	Валовой сбор, тыс. т.	11	14	14	15	15
5	Уборочная площадь, тыс. га	6	10	9	10	10
6	Урожайность, ц/га	46	48	48	50	50
<b>ячмень</b>						
7	Валовой сбор, тыс. т.	16	18	19	20	20
8	Уборочная площадь, тыс. га	11	11	14	15	15
9	Урожайность, ц/га	51	54	54	55	55
<b>кукуруза на зерно</b>						
10	Валовой сбор, тыс. т.	22	23	25	21	25
11	Уборочная площадь, тыс. га	20	16	20	16	16
12	Урожайность, ц/га	56	58	58	56	60
<b>рис</b>						
13	Валовой сбор, тыс. т.	26	26	30	28	29
14	Уборочная площадь, тыс. га	21	25	25	22	23
15	Урожайность, ц/га	62	61	65	65	65
<b>Сахарная свекла</b>						
16	Валовой сбор, тыс. т.	31	32	32	33	35
17	Уборочная площадь, тыс. га	29	30	26	26	28
18	Урожайность, ц/га	66	67	68	69	70
<b>Подсолнечник</b>						
19	Валовой сбор, тыс. т.	37	38	38	36	40
20	Уборочная площадь, тыс. га	35	35	32	31	33
21	Урожайность, ц/га	71	71	74	72	75
<b>Соя</b>						
22	Валовой сбор, тыс. т.	41	42	42	41	45
23	Уборочная площадь, тыс. га	40	36	36	36	38
24	Урожайность, ц/га	76	77	78	77	80

В таблице 8 объектам обучающей выборки соответствуют столбцы с информацией по годам. В каждой группе строк первая строка содержит коды классов, а две последующие – факторов.

### **Выводы**

Предложенные типовая методика и инструментарий, включая программный интерфейс, *обеспечивают высокую степень автоматизации* выполнения первого этапа системно-когнитивного анализа, т.е. когнитивную структуризацию и формализацию задач интеллектуальной обработки данных с применением системы "Эйдос", что позволяет:

- повысить качество семантических информационных моделей;
- достичь существенной экономии труда и времени на синтез семантической информационной модели объекта исследования;
- исследовать различные варианты моделей, оперативно отслеживать динамику предметной области.

### Список литературы

1. Луценко Е.В. Автоматизированный системно-когнитивный анализ в управлении активными объектами (системная теория информации и ее применение в исследовании экономических, социально-психологических, технологических и организационно-технических систем): Монография (научное издание). – Краснодар: КубГАУ, 2002. – 605 с.
2. Луценко Е.В. Теоретические основы и технология адаптивного семантического анализа в поддержке принятия решений (на примере универсальной автоматизированной системы распознавания образов "ЭЙДОС-5.1"). – Краснодар: КЮИ МВД РФ, 1996. – 280 с.
3. Пат. № 940334. РФ. Универсальная автоматизированная система анализа и прогнозирования ситуаций на фондовом рынке "ЭЙДОС-фонд" / Е.В. Луценко (Россия), Б.Х. Шульман (США); Заяв. № 940336. Оpubл. 23.08.94. – 50 с.

4. Пат. № 2003610987 РФ. Автоматизированная система комплексной обработки данных психологического тестирования "ЭЙДОС-Ψ" / Е.В. Луценко (Россия), С.Д. Некрасов (Россия); Заяв. № 2003610511 РФ. Опубл. от 22.04.2003. – 50 с.

5. Пат. № 2003610986 РФ. Универсальная когнитивная аналитическая система "ЭЙДОС" / Е.В. Луценко (Россия); Заяв. № 2003610510 РФ. Опубл. от 22.04.2003. – 50 с.

6. Пат. № 940217. РФ. Универсальная автоматизированная система распознавания образов "ЭЙДОС" / Е.В. Луценко (Россия); Заяв. № 940103. Опубл. 11.05.94. – 50 с.

7. Луценко Е.В. Методика использования репозитария UCІ для оценки качества математических моделей систем искусственного интеллекта // Научный журнал КубГАУ. 2003. № 2. <http://ej.kubagro.ru> .